Math 277
Ordinary Differential Equations

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 2
Direction Fields

Initializations

```
> restart;
  with(DEtools):
```

2.1 Slope Fields.

Examples

Example 2.1.1

Plot the slope field for the equation $\frac{dy}{dx} = x^2 - y$. Superimpose some solution curves on the slope field.

Solution
Use the **DEplot** routine in the **DEtools** package.

```
> deq:=diff(y(x), x)=x^2-y(x);
  DEplot(deq, y(x), x=-3..3, y=-3..3);
```

$$deq := \frac{d}{dx}\, y(x) = x^2 - y(x)$$



Some solution curves can be added by supplying a few initial conditions.

```
> ic:=[seq([y(0)=k], k=-2..3)];
  DEplot(deq, y(x), x=-3..3, ic, y=-3..3, method=classical
```

```
[rk4], linecolor=blue);
```

$ic := [[y(0) = -2], [y(0) = -1], [y(0) = 0], [y(0) = 1], [y(0) = 2], [y(0) = 3]]$



Compare this result to Figure 1.6 (b) on Page 17 of the textbook.

Example 2.1.2

Plot the slope field for the equation $\frac{dy}{dx} = -\frac{y}{x}$. Superimpose some solution curves on the slope field.

Solution

```
> deq:=diff(y(x),x)=-y(x)/x;
  ic:=[seq([y(1)=k], k=[-2, -1, 1, 2])];
```

$$deq := \frac{d}{dx} y(x) = -\frac{y(x)}{x}$$

$$ic := [[y(1) = -2], [y(1) = -1], [y(1) = 1], [y(1) = 2]] \qquad (2.1.2.1)$$

```
> DEplot(deq, y(x), x=-1..2, ic, y=-2..2, linecolor=gold,
  method=classical[rk4]);
```



Example 2.1.3

The slope field for an autonomous equation $\frac{dp}{dt} = f(p)$ depends only on p, not on t. We expect that any horizontal translation of a solution curve is again a solution curve. As an example we plot some a few solution curves of the logistic equation for the population p (in thousands) for a certain species: $\frac{dp}{dt} = p(2-p)$.

Solution

```
> deq:=diff(p(t), t)=p(t)*(2-p(t));
  ic:=[seq([p(0)=k/2], k=0..5)];
```

$$deq := \frac{d}{dt}\, p(t) = p(t)\,(2 - p(t))$$

$$ic := \left[[p(0)=0], \left[p(0)=\frac{1}{2}\right], [p(0)=1], \left[p(0)=\frac{3}{2}\right], [p(0)=2], \left[p(0)=\frac{5}{2}\right]\right] \quad \textbf{(2.1.3.1)}$$

```
> DEplot(deq, p(t), t=-2..2, ic, p=0..3, linecolor=magenta,
  method=classical[rk4]);
```